
Initializing the 8042 Keyboard Controller

STATUS_PORT     EQU     064H
KBD_OUT_BUF     EQU     060H
KBD_INPT_BUF    EQU     060H
KBD_CMD_BUF     EQU     064H
KBD_DATA_BUF    EQU     060H
PORT_A          EQU     060H
PORT_B          EQU     061H
OUT_BUF_FULL    EQU      01H
INPT_BUF_FULL   EQU      02H
SYS_FLAG        EQU      04H
CMD_DATA        EQU      08H
KEYBD_INH       EQU      10H
TRANS_TMOUT     EQU      20H
RCV_TMOUT       EQU      40H
PARITY_EVEN     EQU      80H
INH_KEYBOARD    EQU      10H

KBD_ENA         EQU     0AEH
KBD_DIS         EQU     0ADH

; Keyboard Commands
KB_MENU         EQU     0F1H
KB_ENABLE       EQU     0F4H
KB_MAKEBREAK    EQU     0F7H
KB_ECHO         EQU     0FEH
KB_RESET        EQU     0FFH
KB_LED_CMD      EQU     0EDH
; Keyboard responses
KB_OK           EQU     0AAH
KB_ACK          EQU     0FAH
KB_OVERRUN      EQU     0FFH
KB_RESEND       EQU     0FEH
KB_BREAK        EQU     0F0H
KB_FA           EQU     010H
KB_FE           EQU     020H
KB_PR_LED       EQU     040H



;****************************************************************
;INITKBD
; This procedure intializes the 8042
;****************************************************************

INITKBD     PROC    FAR

            PUSH    AX
            PUSH    DX
            PUSH    CX

            SUB     CX, CX
KBD1:       IN      AL, STATUS_PORT      ; wait for input buffer empty
            TEST    AL, INPT_BUF_FULL
            LOOPNZ  KBD1

            MOV     AL, 0AAH             ;Send self test command
            OUT     KBD_CMD_BUF, AL

            SUB     CX, CX               ; Wait for test to complete
KBD2:       IN      AL, STATUS_PORT
            TEST    AL, OUT_BUF_FULL
            LOOPZ   KBD2

            IN      AL, KBD_OUT_BUF      ;Check self test result
      CMP    AL, 055H

            JNE     KBDF

            MOV     AL, 0ABH             ;Send test Interface command
            OUT     KBD_CMD_BUF, AL

            SUB     CX, CX               ; wait for test to complete
KBD3:       IN      AL, STATUS_PORT
            TEST    AL, OUT_BUF_FULL
            LOOPZ   KBD3

            IN      AL, KBD_OUT_BUF      ;Display Interface test results
      CMP    AL, 000H      ;Should be 00

            JNE     KBDF

            MOV     AL, 060H             ;Send command byte
            OUT     KBD_CMD_BUF, AL
            SUB     CX, CX
KBD4:       IN      AL, STATUS_PORT      ;Wait until input buffer
        TEST    AL, INPT_BUF_FULL  ;is not full
            LOOPNZ  KBD4

            MOV     AL, 01101001B        ;Send command byte
            OUT     KBD_DATA_BUF, AL

            SUB     CX, CX
KBD5:       IN      AL, STATUS_PORT      ;Wait until input buffer is not full
            TEST    AL, INPT_BUF_FULL
            LOOPNZ  KBD5

            MOV     AL, 0FFH             ; Send Keyboard Reset command
            OUT     KBD_DATA_BUF,AL
            CALL    DLY1

            SUB     CX, CX
KBD6:       IN      AL, STATUS_PORT      ;Wait until output buffer is full
            TEST    AL, OUT_BUF_FULL
            LOOPZ   KBD6
            IN      AL, KBD_OUT_BUF

            MOV     AL, 0EEH             ; Send Echo EE Command
            OUT     KBD_DATA_BUF,AL
            CALL    DLY1



            SUB     CX, CX
KBD7:       IN      AL, STATUS_PORT      ;Wait until output buffer is full
            TEST    AL, OUT_BUF_FULL
            LOOPZ   KBD7

            IN      AL, KBD_OUT_BUF      ; Read Echo from Keyboard
      CALL    DLY1
            CMP     AL, 0EEH
            JNE     KBDF

            MOV     AL, 0F4H             ; Enable the Keyboard
            OUT     KBD_DATA_BUF, AL

            SUB     CX, CX
KBD8:       IN      AL, STATUS_PORT      ;Wait until output buffer is full
            TEST    AL, OUT_BUF_FULL
            LOOPZ   KBD8

            IN      AL, KBD_OUT_BUF      ; Clear the output buffer

            POP     CX
            POP     DX
            POP     AX
            RET

KBDF:       MOV     AL, 0FFH
            AL2DPLY
            JMP     $

INITKBD     ENDP

;********************************************
C8042      PROC FAR
           OUT     STATUS_PORT, AL

           SUB     CX, CX
C42:       IN      AL, STATUS_PORT
           TEST    AL, INPT_BUF_FULL
           LOOPNZ  C42
           RET
C8042      ENDP


